Név: Anonymous

1. dolgozat
Számítógéppel segített matematikai modellezés
"J" változat
2012. november 12, kedd

Oldd meg a következő feladatokat. Készíts szép notebook-ot, figyelj a korrekt strukturált megoldásokra, válaszolj a kérdésekre.

---

1. feladat (15 pont)

a) Készítsd el az alábbi táblázat elsô 20 sorát!

a j-edik sorban a j-edik természetes szám áll, utána a j-edik “hatos/ szexi prímtrió” áll, azaz olyan rendezett hármas, melyre (p és p+6, p+12 prím, de p+18 összetett), a 2. sor:

```
2     17 23 29
```

```
In[12]:=
SexyQ[p_] := PrimeQ[p + 6] ∧ PrimeQ[p + 12] ∧ ¬ PrimeQ[p + 18]

In[13]:= ST = Select[Table[Prime[n], {n, 160}], SexyQ]

Out[13]= {7, 17, 31, 47, 67, 97, 101, 151, 167, 227, 257, 271, 347, 367, 557, 587, 607, 647, 727, 941}

Length[%]

20

TableForm[Table[{Style[n, Red], ST⟦n⟧, ST⟦n⟧ + 6, ST⟦n⟧ + 12}, {n, 20}]]
```

1	7	13	19
2	17	23	29
3	31	37	43
4	47	53	59
5	67	73	79
6	97	103	109
7	101	107	113
8	151	157	163
9	167	173	179
10	227	233	239
11	257	263	269
12	271	277	283
13	347	353	359
14	367	373	379
15	557	563	569
16	587	593	599
17	607	613	619
18	647	653	659
19	727	733	739
20	941	947	953

b) Adj meg a k-tökéletes számokat 100000-ig (1≤k≤5), azaz olyan n természetes számokat generálj, melyre ((σ(n) = k*n) ∧ (n<10^5), σ-osztók összege). Találsz-e 4-, ill. 5-perfekt számot? Ha igen, add meg az osztóik öszegét is!

```
Table[Select[Table[j, {j, 10^5}], Plus @@ Divisors[#] == k # &], {k, 5}]
```

```
{{1}, {6, 28, 496, 8128}, {120, 672}, {30240, 32760}, {}}
```

2 4-perfekt szám van 10^5-ig, nincs 5-perfekt.

```
DivisorSigma[1, 30240]
```

```
120960
```

```
DivisorSigma[1, 32760]
```

```
131040
```

c) Add meg az $f(x)$ = arctg(x) fgv. nulla körüli, k-adik $T_k(x)$ Taylor polinomjait (0≤k≤10) . Adj ez alapján képletet $T_n(x)$-re! (n∈ℕ) Ábrázold a polinomokat a [-2,2] intervallumon egy ábrán f-fel!

In[14]:=
```
T = Table[Normal[Series[ArcTan[x], {x, 0, k}]], {k, 0, 10}]
```

Out[14]=

$$\left\{0, x, x, x-\frac{x^3}{3}, x-\frac{x^3}{3}, x-\frac{x^3}{3}+\frac{x^5}{5}, x-\frac{x^3}{3}+\frac{x^5}{5}, x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}, x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}, x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}+\frac{x^9}{9}, x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}+\frac{x^9}{9}\right\}$$

In[15]:=
```
TableForm[T]
```

Out[15]//TableForm=

0

x

x

$x-\frac{x^3}{3}$

$x-\frac{x^3}{3}$

$x-\frac{x^3}{3}+\frac{x^5}{5}$

$x-\frac{x^3}{3}+\frac{x^5}{5}$

$x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}$

$x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}$

$x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}+\frac{x^9}{9}$

$x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}+\frac{x^9}{9}$

```
Plot[Evaluate[Join[{ArcTan[x]}, T]], {x, -2, 2}, PlotStyle → Join[{{Red}}, Table[{Black}, {11}]]]
```



Képlet: $T_{2n+1}(x) = \left(\sum_{k=0}^{n} (-1)^k\, x^{2k+1} / (2k+1)\right)$

---

2. feladat (10 pont)

Ábrázold a

$$g(x) = x \ln^2 |x|$$

függvényt a [-10,10] intervallumon. Mi az értelmezési tartomány?
Add meg $-\infty$-ben a limeszt!

Add meg a lokális szélsôértékeket, színezz (új ábrán) az elsô derivált elôjele szerint!
Ábrázold együtt g és a g' függvényeket is.
Hogyan kell az elsô ábrát módosítani, hogy lokális szélsôértékek vizuálisan is szemléltethetôk legyenek?
Jelöld a lokális minimumokat il. maximumokat az ábrán is.

Add meg a limeszeket a további kritikus pontokban.

In[6]:=
```
g[x_] = x Log[Abs[x]]^2
```

Out[6]= $x \,\mathrm{Log}[\mathrm{Abs}[x]]^2$

In[7]:= `Plot[g[x], {x, -10, 10}]`

Out[7]=



ET=ℝ\ {0}

`Limit[g[x], x → -∞]`

$-\infty$

`Solve[D[g[x], x] == 0, x]`

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

$\{\{x \to 1\}\}$

`Reduce[D[g[x], x] == 0, x, Reals]`

Reduce::fexp :
Warning: Reduce used FunctionExpand to transform the system. Since FunctionExpand transformation rules are
only generically correct, the solution set might have been altered. ≫

$\text{Abs}'[x] \in \text{Reals} \,\&\&\, \left(x == -1 \,||\, x == 1 \,||\, x == -\frac{1}{e^2} \,||\, x == \frac{1}{e^2}\right)$

In[8]:= `g2[x_] := Piecewise[{{0, x == 0}, {x Log[x]^2, x > 0}, {x Log[-x]^2, True}}]`

In[16]:= `x /. Solve[D[g2[x], x] == 0, x]`

Out[16]= $\left\{-1, -\frac{1}{e^2}, 1, \frac{1}{e^2}\right\}$

In[17]:= `lelist = x /. Solve[D[g2[x], x] == 0, x]`

Out[17]= $\left\{-1, -\frac{1}{e^2}, 1, \frac{1}{e^2}\right\}$

```
Plot[g2[x], {x, -2, 2}, ColorFunction → (If[(Evaluate[D[g2[x], x]] /. x → #1) > 0, Red, Blue] &),
 ColorFunctionScaling → False, Epilog → {PointSize[.02], Green, Map[Point[{#, g2[#]}] &, lelist]}]
```



In[11]:=
```
Plot[Evaluate[{g2[x], D[g2[x], x]}], {x, -2, 2},
 ColorFunction → (If[(Evaluate[D[g2[x], x]] /. x → #1) > 0, Red, Blue] &),
 ColorFunctionScaling → False, Epilog → {PointSize[.02], Green, Map[Point[{#, g2[#]}] &, lelist]}]
```

Out[11]=




```
Limit[g[x], x → 0, Direction → 1]
```

0

```
Limit[g[x], x → 0, Direction → -1]
```

0

3. feladat (15 pont)

A numerikus matematikában az $l_j(x) = \prod_{k=1, k\neq j}^{n} (x - x_k) \big/ (x_j - x_k)$ polinomot az XL = $\{x_1, x_2 \ldots, x_n\}$ alappontrendszerhez (elemek pár. kül.) tartozó j-edik interpolációs alappolinomnak nevezik.

a) Add meg azt a függvényt, aminek bemenô paramétere az XL lista és a j index és elôállítja l_j-t.
b) Add meg az l_1, l_2, l_3 polinomot, ha XL={1, 2, 3}.
c) Add meg a $2\,l_1 + 5\,l_2 + 10\,l_3$ polinomot és mutasd meg, hogy grafikonja áthalad a {1, 2}, {2, 5}, {3, 10} pontokon!

Készíts interaktív ábrát, melyben 2 csúszka segítségével ábrázolod az $2\,l_1 - l_2 + 2\,l_3$ függvénycsaládot, ahol

XL={x1,x2=0, x3} x1≠x3, x1,x3∈[-1,1])

Tippek: Sum, Product, Manipulate

In[5]:=
```
BaseInt[j_, XL_List] := Product[(x - XL[[k]]) / (XL[[j]] - XL[[k]]), {k, 1, j - 1}]
  Product[(x - XL[[k]]) / (XL[[j]] - XL[[k]]), {k, j + 1, Length[XL]}]
```

```
BaseInt[1, {1, 2, 3}]
```

$\frac{1}{2}(2-x)(3-x)$

```
Table[BaseInt[j, {1, 2, 3}], {j, 3}]
```

$\left\{\frac{1}{2}(2-x)(3-x),\ (3-x)(-1+x),\ \frac{1}{2}(-2+x)(-1+x)\right\}$

In[18]:=
```
LI = 2 * BaseInt[1, {1, 2, 3}] + 5 * BaseInt[2, {1, 2, 3}] + 10 * BaseInt[3, {1, 2, 3}]
```

Out[18]= $(2-x)(3-x)+5(3-x)(-1+x)+5(-2+x)(-1+x)$

```
Plot[LI, {x, 0, 4}, Epilog → {Red, PointSize[.02], Point[{{1, 2}, {2, 5}, {3, 10}}]}]
```



```
Manipulate[Plot[2*BaseInt[1, {x1, 0, x3}] - BaseInt[2, {x1, 0, x3}] + 2*BaseInt[3, {x1, 0, x3}],
  {x, -1, 1}, PlotRange → {{-1.1, 1.1}, {-6, 6}}], {{x1, -1}, -1, 1, .1}, {{x3, 1}, -1, 1}]
```




```
Manipulate[Plot[2*BaseInt[1, {x1, 0, x3}] - BaseInt[2, {x1, 0, x3}] + 2*BaseInt[3, {x1, 0, x3}],
  {x, -1, 1}, PlotRange → {{-1.1, 1.1}, {-6, 6}},
  Epilog → {Red, PointSize[.02], Point[{{x1, 2}, {0, -1}, {x3, 2}}]}],
 {{x1, -1}, -1, 1, .1}, {{x3, 1}, -1, 1}]
```



4. feladat (15 pont)

Egy szabályos pénzérmét dobálunk. Addig dobáljuk, míg fej nem lesz. Legyen a valószínûségi változónk a dobások száma (1,2,3,...) . Mi a valószínûsége, hogy páros sok dobás után ér véget a dobássorozat?
Készíts kísérletet, melyben a relatív gyakorisággal közelíted a valószínûséget. Végezz 10^3 kísérletet. Vesd össze az értéket a teoretikus valószínûséggel!

*Ha marad idôd, cseréld az érmét (szabályos) kockára és addig dobj, míg 6-ost nem kapsz. Tedd fel ugyanazokat a kérdéseket és végezz ugyanolyan szimulációt, mint az érme esetén!

Tipp: Random[Integer,{0,1}]

```
Random[Integer, {0, 1}]
```

```
1
```

```
In[19]:= M = 1000;
```

Fej→0
Írás→1

```
In[20]:= CoinThCounter[s_Integer] := If[Random[Integer, {0, 1}] == 0, s, CoinThCounter[s + 1]]
```

```
Take[Table[CoinThCounter[1], {M}], 10]
```

```
{1, 1, 4, 1, 1, 2, 1, 1, 1, 1}
```

```
TT = Table[Mod[CoinThCounter[1] + 1, 2], {M}];
```

```
N[Plus @@ TT / M]
```

```
0.331
```

```
1 - %
```

```
0.669
```

Válasz: p=1/3 mivel geometriai eloszlásról van szó

```
In[21]:= p = 1 / 2
```

Out[21]= $\frac{1}{2}$

Sum$\left[p\ (1-p)^{2k-1},\ \{k, 1, \infty\}\right]$

$\frac{1}{3}$

```
In[22]:= CoinThCounterExt[s_Integer, L_List] :=
          If[Random[Integer, {0, 1}] == 0, {s, Append[L, 0]}, CoinThCounterExt[s + 1, Append[L, 1]]]
```

```
TableForm[Table[CoinThCounterExt[1, {}], {10}]]
```

4	1 1 1 0
1	0
1	0
2	1 0
3	1 1 0
1	0
3	1 1 0
2	1 0
1	0
1	0

*kocka:

```
In[23]:= DiceThCounter[s_Integer] := If[Random[Integer, {1, 6}] == 6, s, DiceThCounter[s + 1]]
```

$$TTT = Table\left[Mod[DiceThCounter[1] + 1, 2], \{10^5\}\right];$$

$$N\left[Plus @@ TTT / 10^5\right]$$

0.45452

$$p = 1/6$$

$$\frac{1}{6}$$

$$Sum\left[p\,(1-p)^{2k-1}, \{k, 1, \infty\}\right]$$

$$\frac{5}{11}$$

N[%]

0.454545